Q:

How to draw a "not" colored text?

I'm looking for a way to draw text with inverted colors.
For shapes, we have TPenMode that can be set to pmNot, but we can't do this for text. How can I do this instead?

A:

This does it:
procedure DrawTextNOT(const hDC: HDC; const Font: TFont; const Text: string; const X, Y: integer);
begin
  with TBitmap.Create do
    try
      Canvas.Font.Assign(Font);
      with Canvas.TextExtent(Text) do
        SetSize(cx, cy);
      Canvas.Brush.Color := clBlack;
      Canvas.FillRect(Rect(0, 0, Width, Height));
      Canvas.Font.Color := clWhite;
      Canvas.TextOut(0, 0, Text);
      BitBlt(hDC, X, Y, Width, Height, Canvas.Handle, 0, 0, SRCINVERT);
    finally
      Free;
    end;
end;

Example:
procedure TForm1.FormClick(Sender: TObject);
begin
  Canvas.Brush.Color := clRed;
  Canvas.FillRect(ClientRect);
  DrawTextNOT(Canvas.Handle, Canvas.Font, 'This is a test.', 20, 100);
//  DrawTextNOT(Canvas.Handle, Canvas.Font, 'This is a test.', 20, 100);
end;

You probably also want to disable ClearType. To do that, I refer you to a previous SO question.

